
Change 1
	Testcase name
	9.2.1.1.28

	Reason for change
	Some UEs want to deregister from IMS prior to detach from LTE. Current implementation does not handle this in this TC and it fails. Implemented handling of IMS deregistering on IMS side.

	Summary of change
	Added handling of optional deregistration on IMS side.
Imported all the neccessary modules.

	Source of change
	\LTE\9_2\NAS_Attach_EOnly_IMS.ttcn


Before change:

	module NAS_Attach_EOnly_IMS {

  import from IMS_Component all;

  import from IMS_PTC_CoordMsg all;

  import from IMS_Procedures_Registration all;

  /*

   * @desc      REFERENCE TS 36.523-1 clause 9.2.1.1.28

   * @status    APPROVED (LTE)

   */

  function f_TC_9_2_1_1_28_IMS1() runs on IMS_PTC

  {

    // Preamble   @sic R5s131000 change 1 sic@

    f_IMS_PTC_Init(PDN_1, -, IMS_Security);

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);     // send trigger to EUTRA PTC to indicate start of IMS registration

    f_IMS_Registration();

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);     // send trigger to EUTRA PTC to indicate end of IMS registration

    f_IMS_IPCAN_WaitForTrigger(IPCAN);   // @sic R5s131000 MCC160 implementation: wait for the UE getting switched off sic@
    f_IMS_PTC_Init(PDN_1, -, IMS_Security);

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);     // @sic R5s131000 MCC160 implementation: send trigger to EUTRA PTC to indicate that IMS PTC is reset again sic@

    f_IMS_TestBody_Set(true);

    // Registration: 34.229-1 Annex C.2

    f_IMS_Registration();

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step C.2");

    // Inform EUTRA: IMS Registration done

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_TestBody_Set(false);

  }

}


After change :

	module NAS_Attach_EOnly_IMS {

  import from IMS_Component all;

  import from IMS_PTC_CoordMsg all;

  import from IMS_Procedures_Registration all;

  import from IMS_Component all;

  import from IMS_SIP_Templates all;

  import from IMS_ASP_Templates all;

  import from CommonDefs all; 

  import from IMS_IP_Config all; 

  import from LibSip_SIPTypesAndValues all;

  import from IMS_ASP_TypeDefs all;
  /*

   * @desc      REFERENCE TS 36.523-1 clause 9.2.1.1.28

   * @status    APPROVED (LTE)

   */

  function f_TC_9_2_1_1_28_IMS1() runs on IMS_PTC

  {

    var template (present) SipUrl v_SipUrl;

    var IMS_DATA_REQ v_IMS_DATA_REQ;
    // Preamble   @sic R5s131000 change 1 sic@

    f_IMS_PTC_Init(PDN_1, -, IMS_Security);

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);     // send trigger to EUTRA PTC to indicate start of IMS registration

    f_IMS_Registration();

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);     // send trigger to EUTRA PTC to indicate end of IMS registration

    v_SipUrl := cr_SipUri_HostPort(f_IMS_PTC_ImsInfo_GetHomeDomainName());

    alt { //de-registration is not mandatory

     [] IMS_Server.receive(car_IMS_Register_Request(cr_REGISTER_Request(v_SipUrl))) -> value v_IMS_DATA_REQ

       {

         f_IMS_PTC_Deregistration(v_IMS_DATA_REQ);

         f_Delay(3.0);

         f_IMS_CloseTCP();

         repeat;

       }

     [] IPCAN.receive(cmr_IMS_IPCAN_Trigger)  //Wait for the UE getting switched off
       {

       }

    }
    f_IMS_PTC_Init(PDN_1, -, IMS_Security);

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);     // @sic R5s131000 MCC160 implementation: send trigger to EUTRA PTC to indicate that IMS PTC is reset again sic@

    f_IMS_TestBody_Set(true);

    // Registration: 34.229-1 Annex C.2

    f_IMS_Registration();

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step C.2");

    // Inform EUTRA: IMS Registration done

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_TestBody_Set(false);

    alt { //de-registration is not mandatory

     [] IMS_Server.receive(car_IMS_Register_Request(cr_REGISTER_Request(v_SipUrl))) -> value v_IMS_DATA_REQ

       {

         f_IMS_PTC_Deregistration(v_IMS_DATA_REQ);

         f_Delay(3.0);

         f_IMS_CloseTCP();

         repeat;

       }

     [] IPCAN.receive(cmr_IMS_IPCAN_Trigger)  //Wait for the UE getting switched off
       {

       }

    }
  }

}


Change 2

	Testcase name
	9.2.1.1.28

	Reason for change
	Some UEs want to deregister from IMS prior to detach from LTE. Current implementation does not handle this in this TC and it fails. Implemented handling of IMS deregistration on IMS side.

Attach request in Step 5 should be integrity protected as the UE was previously attached to the network in Preamble.

Current implementation allows a non-conformant UE to pass the TC as the specific content of PDN CONNECTIVITY REQUEST message is checked but does not trigger fail verdict if the check fails.

	Summary of change
	Added handling of optional deregistration on IMS side.

Attach request expected in Step 5 changed to integrity protected.
Modified content checking of PDN connectivity message to triger fail verdict if check fails.

	Source of change
	\LTE\9_2\NAS_Attach_EOnly.ttcn


Before change:

	function f_TC_9_2_1_1_28_EUTRA() runs on EUTRA_PTC

  { /* Attach / Success / IMS */

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c1);

    // Create cell and send out system information

    f_EUTRA_CellConfig_Def(eutra_CellA);

    f_EUTRA_Preamble(eutra_CellA, STATE2_IDLEUPDATE, NORMAL);

    f_EUTRA_UE_Detach_SwitchOffUe(eutra_CellA, RRC_IDLE, -, NORMAL);

    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]);    // @sic R5s131000 MCC160 implementation: send trigger to IMS PTC to indicate the UE being switched off sic@

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);  // @sic R5s131000 MCC160 implementation: wait for the IMS PTC being reset sic@

    f_IP_Handling_Stop(IP, cs_IP_StopPDN(PDN_1));     // @sic R5s131000 MCC160 implementation: reset IP handling for PDN_1 sic@

    f_EUTRA_TestBody_Set(true);

    fl_TC_9_2_1_1_28_Body();

    // Postamble

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_CellA, E2_CONNECTED, NORMAL);

  }

  /*

   * @desc      test body of TC_9_2_1_1_28

   * @status    APPROVED (LTE)

   */

  function fl_TC_9_2_1_1_28_Body() runs on EUTRA_PTC

  {

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType (NORMAL);

    var template (omit) ProtocolConfigOptions v_Pco := omit;

    var NAS_MSG_Indication_Type v_NasInd;

    var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(NORMAL);

    var B3_Type v_PDNType := f_GetPdnType();

    var ProcedureTransactionIdentifier v_EPS_TI := '00'O;

    var NAS_UL_Message_Type v_PdnConnectivityRequest;

    //@siclog "Step 1" siclog@

    // The UE is switched on.

    f_EUTRA_SwitchOnUEandStartIP(eutra_CellA);

    //@siclog "Step 2" siclog@

    // UE broadcasts System Information

    //@siclog "Step 3" siclog@

    // receive RRC Connection Request on SRB0 (CCCH):

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_CellA, cr_EstablishmentCause_Any);

    //@siclog "Step 4" siclog@

    f_EUTRA_RRC_ConnectionSetup_Def (eutra_CellA, -, v_RRC_TI);

    //@siclog "Step 5" siclog@

    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_CellA,

                                                       cr_508_RRCConnectionSetupComplete(v_RRC_TI, ?),

                                                       cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection, // @sic R5s131000 sic@

                                                                                         cr_508_ATTACH_REQUEST(v_AttachType, v_AdditionalUpdateType),

                                                                                         cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, v_PDNType)));

    v_PdnConnectivityRequest := v_NasInd.Pdu.PiggybackedPduList[0].Msg;

    v_EPS_TI := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier;

    if (ispresent(v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions)) {

      v_Pco := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions;

    }

    if (fl_CheckPCOforSIP_SignallingRequest(v_Pco) and (fl_CheckPCOforPCSCF_IPv6(v_Pco) or fl_CheckPCOforPCSCF_IPv4(v_Pco))) {

      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5");

    }

    //@siclog "Step 6-16" siclog@

    // Steps 5 to 15 of the generic test procedure for UE registration (TS 36.508 4.5.2.3-1).

    f_EUTRA_IdleUpdated_Step5_15(eutra_CellA,

                                 STATE2_IDLEUPDATE,

                                 NORMAL,

                                 v_NasInd);

    // EXCEPTION: In parallel to the event described in step 17 below, if initiated by the UE the generic procedure for IP address allocation in the U-plane

    // specified in TS 36.508 subclause 4.5A.1 takes place performing IP address allocation in the U-plane.

    // EXCEPTION: In parallel to the event described in step 16 below the generic procedure for IMS signalling in the U-plane specified in table 9.2.1.1.28.3.2-2 takes place.

    //@siclog "Step 17" siclog@

    f_EUTRA_NAS_AttachComplete(eutra_CellA, tsc_EpsDefaultBearerId, noDelay); // Step 16 of the generic test procedure for UE registration (TS 36.508 4.5.2.3-1).

    //Wait Trigger from IMS

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);

  }


After change :

	{ /* Attach / Success / IMS */

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c1);

    // Create cell and send out system information

    f_EUTRA_CellConfig_Def(eutra_CellA);

    f_EUTRA_Preamble(eutra_CellA, STATE2_IDLEUPDATE, NORMAL);

    f_EUTRA_UE_Detach_SwitchOffUe(eutra_CellA, RRC_IDLE, -, NORMAL);

    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]);    // @sic R5s131000 MCC160 implementation: send trigger to IMS PTC to indicate the UE being switched off sic@

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);  // @sic R5s131000 MCC160 implementation: wait for the IMS PTC being reset sic@

    f_IP_Handling_Stop(IP, cs_IP_StopPDN(PDN_1));     // @sic R5s131000 MCC160 implementation: reset IP handling for PDN_1 sic@

    f_EUTRA_TestBody_Set(true);

    fl_TC_9_2_1_1_28_Body();

    // Postamble

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_CellA, E2_CONNECTED, NORMAL);

    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]);  //UE is switched off
  }

  /*

   * @desc      test body of TC_9_2_1_1_28

   * @status    APPROVED (LTE)

   */

  function fl_TC_9_2_1_1_28_Body() runs on EUTRA_PTC

  {

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType (NORMAL);

    var template (omit) ProtocolConfigOptions v_Pco := omit;

    var NAS_MSG_Indication_Type v_NasInd;

    var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(NORMAL);

    var B3_Type v_PDNType := f_GetPdnType();

    var ProcedureTransactionIdentifier v_EPS_TI := '00'O;

    var NAS_UL_Message_Type v_PdnConnectivityRequest;

    //@siclog "Step 1" siclog@

    // The UE is switched on.

    f_EUTRA_SwitchOnUEandStartIP(eutra_CellA);

    //@siclog "Step 2" siclog@

    // UE broadcasts System Information

    //@siclog "Step 3" siclog@

    // receive RRC Connection Request on SRB0 (CCCH):

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_CellA, cr_EstablishmentCause_Any);

    //@siclog "Step 4" siclog@

    f_EUTRA_RRC_ConnectionSetup_Def (eutra_CellA, -, v_RRC_TI);

    //@siclog "Step 5" siclog@

        v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_CellA,

                                                       cr_508_RRCConnectionSetupComplete(v_RRC_TI, ?),

                                                       cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected, // @sic R5s131000 sic@

                                                                                         cr_508_ATTACH_REQUEST(v_AttachType, v_AdditionalUpdateType),

                                                                                         cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, v_PDNType)));
    v_PdnConnectivityRequest := v_NasInd.Pdu.PiggybackedPduList[0].Msg;

    v_EPS_TI := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier;

    if (ispresent(v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions)) {

      v_Pco := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions;

    }

    if (fl_CheckPCOforSIP_SignallingRequest(v_Pco) and (fl_CheckPCOforPCSCF_IPv6(v_Pco) or fl_CheckPCOforPCSCF_IPv4(v_Pco))) {

      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5");

    }

    else {

      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid PCO element in PDN connectivity request message in Step 5");
    }
    //@siclog "Step 6-16" siclog@

    // Steps 5 to 15 of the generic test procedure for UE registration (TS 36.508 4.5.2.3-1).

    f_EUTRA_IdleUpdated_Step5_15(eutra_CellA,

                                 STATE2_IDLEUPDATE,

                                 NORMAL,

                                 v_NasInd);

    // EXCEPTION: In parallel to the event described in step 17 below, if initiated by the UE the generic procedure for IP address allocation in the U-plane

    // specified in TS 36.508 subclause 4.5A.1 takes place performing IP address allocation in the U-plane.

    // EXCEPTION: In parallel to the event described in step 16 below the generic procedure for IMS signalling in the U-plane specified in table 9.2.1.1.28.3.2-2 takes place.

    //@siclog "Step 17" siclog@

    f_EUTRA_NAS_AttachComplete(eutra_CellA, tsc_EpsDefaultBearerId, noDelay); // Step 16 of the generic test procedure for UE registration (TS 36.508 4.5.2.3-1).

    //Wait Trigger from IMS               

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);

  }

