
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 6.1.1.7. Only essential changes to the TTCN are applied and documented in section 4.

With these changes applied the test case can be demonstrated to run with one or more LTE UEs 
(see section 5). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
TC_6_1_1_7
Test Group:
LTE_A//6_1/Idle_PLMNSelection_A.ttcn
ATS Version:
13wk49+ essential modifications.

System Simulator used:
Rohde & Schwarz CMW500
UE used:
Intel XMM 7260
Verification Status:
PASS
4 Corrections required for test case 6.1.1.7
4.1 f_TC_6_1_1_7_EUTRA
	Object name 
	f_TC_6_1_1_7_EUTRA

	Reason for change
	1. The local varialbes “v_KsiValue, v_GutiParams, v_PLMN, v_LAC, v_Tac, v_LAI, v_TaiList are initialised before the creation of cells and would results in invalid values.
2. The timer “tsc_IdleMode_GenericTimer” is not used in the test case and can be removed.

3. After SWITCH ON, the t_HpPlmn_lower and t_HpPlmn_upper are started. We need to ensure that local confirmation is received before actually starting the timer.

4. The timers are protocol timers and therefore the tolerance values needs to be taken into consideration.

5. At Step 20, once the lower timer is expired and RRC Connection request is received, the upper timer needs to be stopped.

6. Incorrect Cell Id is referenced in the comment section for Step 20
7. According to the prose the “t_HpPlmn_upper” timer should be started at Step 20 , to check perform the check at step 28

	Summary of change
	1. Variables are assigned after cell creation and when required.
2. Commented out tsc_IdleMode_GenericTimer

3. Added “CNF_REQUIRED” to f_EUTRA_SwitchOnUEandStartIP  at step 20 to expect Confirmation.

4. Changed the timers to nasTimer

5. Included t_HpPlmn_upper.stop  after receiving RRC connection request message.

6. Changed the comment to indicate it should be Cell 13

7. Moved the start of “t_HpPlmn_upper” from step 24 to Step 20 and also set the timer as nasTimer

	Module
	Idle_PLMNSelection_A.ttcn


Before change: 

	...

function f_TC_6_1_1_7_EUTRA ( ) runs on EUTRA_PTC
  { /* PLMN selection / Periodic reselection / MinimumPeriodicSearchTimer */
    var template (value) CellPowerList_Type v_CellPowerList_AtT1, v_CellPowerList_AtT2;
    var EUTRA_CellId_Type v_CellId := eutra_Cell13;
    var EUTRA_ATTACH_Type v_ForcedAttach := NORMAL;
    var NAS_MSG_Indication_Type v_NasInd;
    var NAS_KsiValue v_KsiValue := f_EUTRA_SecurityKSIasme_Get();
    var GutiParameters_Type v_GutiParams:= f_EUTRA_CellInfo_GetGuti(v_CellId);
    var B3_Type v_EpsUpdate_TypeValue := f_GetEPSTAUType (v_ForcedAttach);
    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus;
    var template AdditionalUpdateType v_AdditionalUpdateType_Expected := f_GetAdditionalUpdateType (v_ForcedAttach);
    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity);
    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode(v_CellId);
    var TrackingAreaCode v_Tac := f_EUTRA_CellInfo_GetTAC(v_CellId);
    var template (omit)  LocAreaId v_LAI := f_GetLAI(v_PLMN, v_LAC, v_ForcedAttach);
    var template (value) TrackingAreaIdList v_TaiList := cds_TAIListNonConsecutive_tlv(v_PLMN, { bit2oct(v_Tac) });
    var template (omit)  MobileIdentity v_MSId := f_GetMSId(v_ForcedAttach);
    timer t_HpPlmn_lower;
    timer t_HpPlmn_upper;
    timer t_IdleMode_GenericTimer := tsc_IdleMode_GenericTimer;
    v_CellPowerList_AtT1 := {
      cs_CellPower ( eutra_Cell13, tsc_ServingCellRS_EPRE )
    };
    v_CellPowerList_AtT2 := {
      cs_CellPower ( eutra_Cell1, tsc_ServingCellRS_EPRE )
    };
    //Initialise all cell, security and mobile parameters
    f_EUTRA_Init(c1);//SIB2 & SIB3 are transmitted
    //Set Sys info and cell info for cell 12, &13 to be used @ creation
    f_EUTRA_CellInfo_SetPLMN_1Entry ( eutra_Cell12, cs_HPLMN_001_11 );  //PLMN1
    f_EUTRA_CellInfo_SetPLMN_1Entry ( eutra_Cell13, cs_HPLMN_001_21 );  //PLMN2
    //Set TAC for Cell1, Cell12, &Cell13
    //TAC for Cell1 is set to 1 in f_EUTRA_Init
    f_EUTRA_CellInfo_SetTAC ( eutra_Cell12, tsc_IdleMode_TAC_Cell12 );
    f_EUTRA_CellInfo_SetTAC ( eutra_Cell13, tsc_IdleMode_TAC_Cell13 );
    //Configure and activate all cells
    //Cell power level will be -85 dBm/15kHz & attenuation set so that cell is seen as OFF
    f_EUTRA_CellConfig_Def ( eutra_Cell12 );
    f_EUTRA_CellConfig_Def ( eutra_Cell1 );
    f_EUTRA_CellConfig_Def ( eutra_Cell13 );
    //Equip UE with a USIM containing default values except for those shown in Table 6.1.1.7.3.1-2
    f_UT_USIM_Insert ( UT, "36.523-1 Table 6.1.1.7.3.1-2" );
    //Preamble to ensure UE is in known state
    f_EUTRA_Preamble ( eutra_Cell12, STATE2_IDLEUPDATE );
    f_EUTRA_UE_Detach_SwitchOffUe ( eutra_Cell12, RRC_IDLE );
    f_EUTRA_TestBody_Set ( true );
    //@siclog "Step 1" siclog@
    //Adjust cell configuration according to T1
    f_EUTRA_SetCellPowerList ( v_CellPowerList_AtT1 );
    //@siclog "Step 2" siclog@
    //Power on UE
    f_EUTRA_SwitchOnUEandStartIP(eutra_Cell12);
    t_HpPlmn_lower.start ( f_EUTRA_SetTimerToleranceMin ( eutra_Cell12, nonProtocolTimer, 120.0 ) );
    t_HpPlmn_upper.start ( f_EUTRA_SetTimerToleranceMax ( eutra_Cell12, nonProtocolTimer, 420.0 ) ) ;
    t_IdleMode_GenericTimer.start;
    //@siclog "Step 3-18" siclog@
    f_EUTRA_IdleUpdated(eutra_Cell12, PREAMBLE, STATE2_IDLEUPDATE,  NORMAL);
    //@siclog "Step 19" siclog@
    f_UT_PLMN_Check(UT, "00111");
    //@siclog "Step 20" siclog@
    //Check for RRCConnectionRequest on Cell1 / PLMN4
    alt {
      [] SRB.receive( car_SRB0_RrcPdu_IND ( eutra_Cell13, cr_508_RRCConnectionRequest ) )
        {
          //* @verdict fail RRCConnectionRequest received on PLMNN2 / Cell 13 within t_HpPlmn_lower time
          f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 6.1.1.7 Step 20");
        }
      [] t_HpPlmn_lower.timeout
        {
          if (f_EUTRA_RRC_ConnectionRequest_Common(eutra_Cell13, t_HpPlmn_upper)) {
            //* @verdict pass RRC Connection request received on PLMN2 / Cell 13 after t_HpPlmn_lower time but before t_HpPlmn_upper
            f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.7 Step 20");
          }
          else {
            //* @verdict fail No RRCConnectionRequest received
            f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 6.1.1.7 Step 20");
          }
        }
    }
    //@siclog "Step 21 - 25" siclog@
    //Perform TAU procedure steps 2 to 6
    //Step 21: Send RRC Connection Setup on SRB0 (CCCH) and switch on UL grant assignments:
    f_EUTRA_RRC_ConnectionSetup_Def(v_CellId, -, tsc_RRC_TI_Def);
    // UE shall establish SRB1 acc. to 36.331 cl. 5.3.3.4 and 5.3.9
    // UE enters RRC_CONNECTED state
    // SRB1 is established: no ciphering, no integrity protection for NAS/RRC
    //Step 22: Receive RRCConnectionSetupComplete which includes a TRACKING AREA UPDATE REQUEST message
    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(v_CellId,
                                                       cr_508_RRCConnectionSetupComplete (tsc_RRC_TI_Def, ?),
                                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                                                         cr_508_TAU_Request(v_EpsUpdate_TypeValue,
                                                                                            v_KsiValue,
                                                                                            cr_DRXparameter_Any ('5C'O) ifpresent,
                                                                                            v_AdditionalUpdateType_Expected)));
    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept
    /* Step 23: send TAU ACCEPT and receive TAU COMPLETE; */
    SRB.send(cas_SRB1_NasPdu_REQ(v_CellId,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                cs_508_TAU_Accept(v_EpsUpdate_TypeValue,
                                                                  f_GutiParameters2MobileIdentity (tsc_IEI_Guti, v_GutiParams),
                                                                  v_TaiList,
                                                                  v_EPS_ContextStatus,
                                                                  v_LAI,
                                                                  v_MSId,
                                                                  f_GetAdditionalUpdateResult(f_GetAdditionalUpdateType (v_ForcedAttach), v_ForcedAttach)))));
    t_HpPlmn_upper.start ( f_EUTRA_SetTimerToleranceMax ( eutra_Cell1, nonProtocolTimer, 420.0 ) ) ;
    /* Step24: receive TAU COMPLETE */
    SRB.receive(car_SRB1_NasPdu_IND(v_CellId,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_508_TAU_Complete)));
    //Step 25: Release RRC connection
    f_EUTRA_RRC_ConnectionRelease(v_CellId, tsc_RRC_TI_Def);
    //@siclog "Step 26" siclog@
    //Check if PLMN2 is indicated by UE
    f_UT_PLMN_Check ( UT, "00121" );
    //* @verdict pass PLMN2 is indicated by UE
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.7 Step 26");
    //@siclog "Step 27" siclog@
    //Adjust cell power level according to T2
    f_EUTRA_SetCellPowerList ( v_CellPowerList_AtT2 );
    //@siclog "Step 28" siclog@
    //Check for RRCConnectionRequest on Cell1 / PLMN4
    alt {
      [] SRB.receive( car_SRB0_RrcPdu_IND ( eutra_Cell1, cr_508_RRCConnectionRequest ) )
        {
          //* @verdict fail RRCConnectionRequest received on PLMNN4 / Cell 1 within t_HpPlmn_Upper time
          f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 6.1.1.7 Step 28");
        }
      [] t_HpPlmn_upper.timeout
        {
          f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1);
          //* @verdict pass RRC Connection request received on PLMN4 / Cell 1 after t_HpPlmn_Upper time
          f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.7 Step 28");
        }
    }
    //@siclog "Step 29 - 33" siclog@
    //Perform TAU procedure steps 2 to 6
    f_EUTRA_TrackingAreaUpdate_WithoutRrcConnReq ( eutra_Cell1 );
    //@siclog "Step 34" siclog@
    //Check if PLMN4 is indicated by UE
    f_UT_PLMN_Check ( UT, "00101" );
    //* @verdict pass PLMN4 is indicated by UE
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.7 Step 34");
    f_EUTRA_TestBody_Set ( false );
    //Switch/power off UE
    f_EUTRA_Postamble ( eutra_Cell1, E1_IDLE );
  } //end of f_TC_6_1_1_7_EUTRA
...


After change: (highlight the changes in Yellow)
	...

function f_TC_6_1_1_7_EUTRA ( ) runs on EUTRA_PTC
  { /* PLMN selection / Periodic reselection / MinimumPeriodicSearchTimer */
    var template (value) CellPowerList_Type v_CellPowerList_AtT1, v_CellPowerList_AtT2;
    var EUTRA_CellId_Type v_CellId := eutra_Cell13;
    var EUTRA_ATTACH_Type v_ForcedAttach := NORMAL;
    var NAS_MSG_Indication_Type v_NasInd;
    var NAS_KsiValue v_KsiValue;
    var GutiParameters_Type v_GutiParams ;
    var B3_Type v_EpsUpdate_TypeValue := f_GetEPSTAUType (v_ForcedAttach);
    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus;
    var template AdditionalUpdateType v_AdditionalUpdateType_Expected := f_GetAdditionalUpdateType (v_ForcedAttach);
    var NAS_PlmnId v_PLMN;
    var NAS_Lac v_LAC;
    var TrackingAreaCode v_Tac;
    var template (omit)  LocAreaId v_LAI;
    var template (value) TrackingAreaIdList v_TaiList;
    var template (omit)  MobileIdentity v_MSId := f_GetMSId(v_ForcedAttach);
    timer t_HpPlmn_lower;
    timer t_HpPlmn_upper;
  //  timer t_IdleMode_GenericTimer := tsc_IdleMode_GenericTimer;
    v_CellPowerList_AtT1 := {
      cs_CellPower ( eutra_Cell13, tsc_ServingCellRS_EPRE )
    };
    v_CellPowerList_AtT2 := {
      cs_CellPower ( eutra_Cell1, tsc_ServingCellRS_EPRE )
    };
    //Initialise all cell, security and mobile parameters
    f_EUTRA_Init(c1);//SIB2 & SIB3 are transmitted
    //Set Sys info and cell info for cell 12, &13 to be used @ creation
    f_EUTRA_CellInfo_SetPLMN_1Entry ( eutra_Cell12, cs_HPLMN_001_11 );  //PLMN1
    f_EUTRA_CellInfo_SetPLMN_1Entry ( eutra_Cell13, cs_HPLMN_001_21 );  //PLMN2
    //Set TAC for Cell1, Cell12, &Cell13
    //TAC for Cell1 is set to 1 in f_EUTRA_Init
    f_EUTRA_CellInfo_SetTAC ( eutra_Cell12, tsc_IdleMode_TAC_Cell12 );
    f_EUTRA_CellInfo_SetTAC ( eutra_Cell13, tsc_IdleMode_TAC_Cell13 );
    //Configure and activate all cells
    //Cell power level will be -85 dBm/15kHz & attenuation set so that cell is seen as OFF
    f_EUTRA_CellConfig_Def ( eutra_Cell12 );
    f_EUTRA_CellConfig_Def ( eutra_Cell1 );
    f_EUTRA_CellConfig_Def ( eutra_Cell13 );
   // 
    v_GutiParams:= f_EUTRA_CellInfo_GetGuti(v_CellId);
    v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode(v_CellId);
    v_Tac := f_EUTRA_CellInfo_GetTAC(v_CellId);
    v_PLMN := f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity);
    v_LAI := f_GetLAI(v_PLMN, v_LAC, v_ForcedAttach);
    v_TaiList := cds_TAIListNonConsecutive_tlv(v_PLMN, { bit2oct(v_Tac) });
    //Equip UE with a USIM containing default values except for those shown in Table 6.1.1.7.3.1-2
    f_UT_USIM_Insert ( UT, "36.523-1 Table 6.1.1.7.3.1-2" );
    //Preamble to ensure UE is in known state
    f_EUTRA_Preamble ( eutra_Cell12, STATE2_IDLEUPDATE );
    f_EUTRA_UE_Detach_SwitchOffUe ( eutra_Cell12, RRC_IDLE );
    f_EUTRA_TestBody_Set ( true );
    //@siclog "Step 1" siclog@
    //Adjust cell configuration according to T1
    f_EUTRA_SetCellPowerList ( v_CellPowerList_AtT1 );
    //@siclog "Step 2" siclog@
    //Power on UE
    f_EUTRA_SwitchOnUEandStartIP(eutra_Cell12,CNF_REQUIRED);
    t_HpPlmn_lower.start ( f_EUTRA_SetTimerToleranceMin ( eutra_Cell12, nasTimer, 120.0 ) );
    t_HpPlmn_upper.start ( f_EUTRA_SetTimerToleranceMax ( eutra_Cell12, nasTimer, 420.0 ) ) ;
  //  t_IdleMode_GenericTimer.start;
    //@siclog "Step 3-18" siclog@
    f_EUTRA_IdleUpdated(eutra_Cell12, PREAMBLE, STATE2_IDLEUPDATE,  NORMAL);
    //@siclog "Step 19" siclog@
    f_UT_PLMN_Check(UT, "00111");
    //@siclog "Step 20" siclog@
    //Check for RRCConnectionRequest on Cell13 / PLMN4
    alt {
      [] SRB.receive( car_SRB0_RrcPdu_IND ( eutra_Cell13, cr_508_RRCConnectionRequest ) )
        {
          //* @verdict fail RRCConnectionRequest received on PLMNN2 / Cell 13 within t_HpPlmn_lower time
          f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 6.1.1.7 Step 20");
        }
      [] t_HpPlmn_lower.timeout
        {
          if (f_EUTRA_RRC_ConnectionRequest_Common(eutra_Cell13, t_HpPlmn_upper)) {
            //* @verdict pass RRC Connection request received on PLMN2 / Cell 13 after t_HpPlmn_lower time but before t_HpPlmn_upper
            f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.7 Step 20");



t_HpPlmn_upper.stop;
          }
          else {
            //* @verdict fail No RRCConnectionRequest received
            f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 6.1.1.7 Step 20");
          }
        }
    }

// 

t_HpPlmn_upper.start ( f_EUTRA_SetTimerToleranceMin ( eutra_Cell13, nasTimer, 420.0 ) ) ;
    //@siclog "Step 21 - 25" siclog@
    //Perform TAU procedure steps 2 to 6
    //
    v_KsiValue := f_EUTRA_SecurityKSIasme_Get();
    //Step 21: Send RRC Connection Setup on SRB0 (CCCH) and switch on UL grant assignments:
    f_EUTRA_RRC_ConnectionSetup_Def(v_CellId, -, tsc_RRC_TI_Def);
    // UE shall establish SRB1 acc. to 36.331 cl. 5.3.3.4 and 5.3.9
    // UE enters RRC_CONNECTED state
    // SRB1 is established: no ciphering, no integrity protection for NAS/RRC
    //Step 22: Receive RRCConnectionSetupComplete which includes a TRACKING AREA UPDATE REQUEST message
    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(v_CellId,
                                                       cr_508_RRCConnectionSetupComplete (tsc_RRC_TI_Def, ?),
                                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                                                         cr_508_TAU_Request(v_EpsUpdate_TypeValue,
                                                                                            v_KsiValue,
                                                                                            cr_DRXparameter_Any ('5C'O) ifpresent,
                                                                                            v_AdditionalUpdateType_Expected)));
    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept
    /* Step 23: send TAU ACCEPT and receive TAU COMPLETE; */
    SRB.send(cas_SRB1_NasPdu_REQ(v_CellId,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                cs_508_TAU_Accept(v_EpsUpdate_TypeValue,
                                                                  f_GutiParameters2MobileIdentity (tsc_IEI_Guti, v_GutiParams),
                                                                  v_TaiList,
                                                                  v_EPS_ContextStatus,
                                                                  v_LAI,
                                                                  v_MSId,
                                                                  f_GetAdditionalUpdateResult(f_GetAdditionalUpdateType (v_ForcedAttach), v_ForcedAttach)))));
    /* Step24: receive TAU COMPLETE */
    SRB.receive(car_SRB1_NasPdu_IND(v_CellId,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_508_TAU_Complete)));
    //Step 25: Release RRC connection
    f_EUTRA_RRC_ConnectionRelease(v_CellId, tsc_RRC_TI_Def);
    //@siclog "Step 26" siclog@
    //Check if PLMN2 is indicated by UE
    f_UT_PLMN_Check ( UT, "00121" );
    //* @verdict pass PLMN2 is indicated by UE
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.7 Step 26");
    //@siclog "Step 27" siclog@
    //Adjust cell power level according to T2
    f_EUTRA_SetCellPowerList ( v_CellPowerList_AtT2 );
    //@siclog "Step 28" siclog@
    //Check for RRCConnectionRequest on Cell1 / PLMN4
    alt {
      [] SRB.receive( car_SRB0_RrcPdu_IND ( eutra_Cell1, cr_508_RRCConnectionRequest ) )
        {
          //* @verdict fail RRCConnectionRequest received on PLMNN4 / Cell 1 within t_HpPlmn_Upper time
          f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 6.1.1.7 Step 28");
        }
      [] t_HpPlmn_upper.timeout
        {
          f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1);
          //* @verdict pass RRC Connection request received on PLMN4 / Cell 1 after t_HpPlmn_Upper time
          f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.7 Step 28");
        }
    }
    //@siclog "Step 29 - 33" siclog@
    //Perform TAU procedure steps 2 to 6
    f_EUTRA_TrackingAreaUpdate_WithoutRrcConnReq ( eutra_Cell1 );
    //@siclog "Step 34" siclog@
    //Check if PLMN4 is indicated by UE
    f_UT_PLMN_Check ( UT, "00101" );
    //* @verdict pass PLMN4 is indicated by UE
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.7 Step 34");
    f_EUTRA_TestBody_Set ( false );
    //Switch/power off UE
    f_EUTRA_Postamble ( eutra_Cell1, E1_IDLE );
  } //end of f_TC_6_1_1_7_EUTRA
...




5 Execution Log Files
5.1 Intel XMM 7260 UE 
The Intel XMM 7260 UE passed this test case on Rohde & Schwarz CMW500 in LTE FDD Band 4. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files: 
tc_6_1_1_7_Intel_XMM_7260.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log provided, masking is applied to prevent any disclosure of confidential information.

In the log file (in .txt format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
6 References

	[1]
	R5s140196
Supporting information for agreement of this test case. 
This archive comprises: 


-  text format execution log files.

