
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 8.3.4.5 in FDD branch which is part of the LTE test suite ‘iwd-EUTRA-B2013-03_D14wk10’.

The test case can be demonstrated to run with one LTE UE(see section 5). Execution logs are provided as evidence.

4) Corrections required for test case 8.3.4.5
Introduction

This section describes the changes required to make test 8.3.4.5 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-B2013-03_D14wk10 release.

4.1 Change 1
	Function name
	f_TC_8_3_4_5_EUTRA

	Reason for change
	1. Initial reference power level of cell 23 needs to be set to highest used value (-79dBm in T1).
2. The pre-test conditions requirement in the 36.523-1 cl. 8.3.4.5.3.1 is that UE’s Allowed CSG list contains the CSG ID of Cell 23. 

It has proved as a good practice for example in the test 6.3.4, based on the PICS settings to clear the Allowed CSG list on the inserted USIM followed by manual selection of the desired cellID - all this as a part of the initial procedure before the test body starts.
3. As approved in RAN4 CR: R4-135104 (respectively R4-140707 which corrects section numbers) the references for testing “leaving” proximity were removed from the 36.133 so there is currently no mechanism to test it. For this reason steps 88-91 as well as T5 can be removed.
Prose CR has been raised for this change.

4. Based on the E-UTRAN measurement procedure for CSG Proximity Indication Testing as described in 36.133, annex A.8.21 the procedure for positive test is following:

Initial condition (A.8.21-1) is to have only Cell1 active and UE is RRC connected there (corresponds to step 83 in current test prose).
Time duration T3 (A.8.21-1) is to turn on CSG cell first (currently corresponds to step 86) and then to send proximityConfig to the UE (currently corresponds to steps 84-85). UE should respond with “entering” proximity (currently corresponds to step 87).

So in order to have the same step sequence in the test as it is given in 36.133, the current step 86 (cells power level set to T4) must be placed after step 83 and before steps 84-85.

Prose CR has been raised for this change.

5. Due to the above described change of removing steps 88-91, the cell on which UE ends the test will newly be cell1 and not cell2.

	Summary of change
	1. Initial reference power level of cell 23 has been set to -79dBm.
2. Based on the PICS settings, the Allowed CSG list will be cleared as part of the test initialisation and CSG Id of cell23 manually selected, if chosen so.

3. Cell power list for T5 and steps 88-91 have been removed. 
4. Current step 86 has been placed between current step 83 and steps 84-85. Also step numbering has been adjusted accordingly 
5. The cell for the postamble has been changed from 2 to 1.

	TTCN module
	LTE\8_3\RRC_Measurement.ttcn


Before change

  function f_TC_8_3_4_5_EUTRA() runs on EUTRA_PTC

  { //Inter-frequency E-UTRAN FDD - FDD / CSG Proximity Indication

    var template (value) CellPowerList_Type v_CellPowerList_T0, v_CellPowerList_T1, v_CellPowerList_T2, v_CellPowerList_T3,v_CellPowerList_T4, v_CellPowerList_T5;

    var CSG_Identity v_CSG_Identity_Cell23 := '000000000000000000000001011'B;

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var GutiParameters_Type v_GutiParams;

    var template (value) MobilityControlInfo v_MobilityControlInfo;

    var template (value) RadioResourceConfigDedicated v_RadioResourceConfigDedicated;

    var template (value) RachProcedureConfig_Type v_RachProcedureConfig;

    var template (value) DL_DCCH_Message v_RRCConnectionReconfiguration;

    var Frequency_IE_Type v_Frequency_IE_Cell23;

    timer t_Timer;

    v_CellPowerList_T0 := {

      cs_CellPower (eutra_Cell1, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_T1 := {

      cs_CellPower (eutra_Cell23, -79)

    };

    v_CellPowerList_T2 := {

      cs_CellPower (eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_ServingCellRS_EPRE),

      cs_CellPower (eutra_Cell23, tsc_NonSuitableOffCellRS_EPRE)

    };

    v_CellPowerList_T3 := {

      cs_CellPower (eutra_Cell1, tsc_ServingCellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE)

    };

    v_CellPowerList_T4 := {

      cs_CellPower (eutra_Cell23, -79)

    };

    v_CellPowerList_T5 := {

      cs_CellPower (eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_ServingCellRS_EPRE),

      cs_CellPower (eutra_Cell23, tsc_NonSuitableOffCellRS_EPRE)

    };
    // Init variables

    f_EUTRA_Init(c3);

    f_EUTRA_ChangeSysinfoCombination(eutra_Cell23, c7);

    // Set SysInfo for cell 23

    f_EUTRA_CellInfo_SetSysInfo_Csg_Ind(eutra_Cell23, true);

    f_EUTRA_CellInfo_SetSysInfo_Csg_Id(eutra_Cell23, v_CSG_Identity_Cell23);

    f_EUTRA_CellInfo_SetSIB4 (eutra_Cell23, cs_SIB4_AllParameters (omit, omit, cs_PhysCellIdRange(2, n4)));

    v_Frequency_IE_Cell23 := f_EUTRA_CellInfo_GetFrequencyIEs (eutra_Cell23);

    // Create and configure Cells

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell2);

    f_EUTRA_CellConfig_Def(eutra_Cell23);

    // Insert a USIM with the Allowed CSG List empty

    f_UT_USIM_Insert(UT, "The UE's Allowed CSG list contains the CSG ID of Cell 23");
    f_EUTRA_TestBody_Set (true);

    // @siclog "Step 1" siclog@
...

   //@siclog "Step 76 - 83" siclog@

    //Steps 2 to 9 of the generic test procedure in TS 36.508 subclause 4.5.3.3 are performed on Cell 1.

    //NOTE: The UE performs the establishment of the new data radio bearer associated with the default EPS bearer context.

    f_EUTRA_RbEst_Def(eutra_Cell1);

    //@siclog "Step 84 - 85" siclog@

    //The SS transmits an RRCConnectionReconfiguration message on Cell 1 including proximityIndicationEUTRA-r9 set to enabled.

    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,

                                                    cs_RRCConnectionReconfiguration_Common(v_RRC_TI,omit,omit,omit,omit,

                                                                                           cs_RRCConnectionReconfiguration_ReportProximityConfig_IEs(enabled)));

    //@siclog "Step 86" siclog@

    //The SS adjusts cell levels according to row T4 of Table 8.3.4.5.3.2-1

    f_EUTRA_SetCellPowerList (v_CellPowerList_T4);
    //@siclog "Step 87" siclog@

    //Check: Does the UE transmit a ProximityIndication message with type-r9 set to "entering" proximity on Cell 1 within  [360]s?

    t_Timer.start(360.0);

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_508_ProximityIndication_Eutra(entering,v_Frequency_IE_Cell23.UL_DL_Earfcn.dl_CarrierFreq)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.3.4.5 Step 87");

    t_Timer.stop;

    //@siclog "Step 88" siclog@

    //The SS adjusts cell levels according to row T5 of Table 8.3.4.5.3.2-1

    f_EUTRA_SetCellPowerList (v_CellPowerList_T5);

    //@siclog "Step 89 - 90" siclog@

    //The SS transmits an RRCConnectionReconfiguration message to order the UE to perform intra frequency handover to Cell 2 and 2 including proximityIndicationEUTRA-r9 set to enabled.

    //The UE transmits an RRCConnectionReconfigurationComplete message on Cell 2 to confirm the successful completion of the intra frequency handover.

    v_MobilityControlInfo := f_Generate_cs_MobilityControlInfo_HO_CRNTI_RACH (eutra_Cell2,omit, tsc_C_RNTI_Def2, cs_RACH_ConfigDedicated);

    v_RadioResourceConfigDedicated := f_Generate_cs_508_RadioResourceConfigDedicated_HO (eutra_Cell2, n64);

    v_RachProcedureConfig := cs_RachProcedureConfig_CRNTI_HO(f_EUTRA_CellInfo_GetRAR_TA(eutra_Cell2),

                                                             f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell2));

    v_RRCConnectionReconfiguration := cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                             omit,

                                                                             v_MobilityControlInfo,

                                                                             v_RadioResourceConfigDedicated,

                                                                             cs_508SecurityConfigHO_IntraLTE,

                                                                             cs_RRCConnectionReconfiguration_ReportProximityConfig_IEs(enabled));

   f_EUTRA_508RRC_IntraLTE_HO_InterCell_Common(eutra_Cell1, eutra_Cell2, v_RRCConnectionReconfiguration, v_RachProcedureConfig);

    //@siclog "Step 91" siclog@

    //Check: Does the UE transmit a ProximityIndication message with type-r9 set to "leaving" proximity on Cell 2 within [360]s?

    t_Timer.start(360.0);

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell2, cr_508_ProximityIndication_Eutra(leaving,v_Frequency_IE_Cell23.UL_DL_Earfcn.dl_CarrierFreq)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.3.4.5 Step 91");

    t_Timer.stop;
    f_EUTRA_TestBody_Set (false);

    // Postamble

    f_EUTRA_Postamble(eutra_Cell2, E2_CONNECTED);

  } // end of f_TC_8_3_4_5_EUTRA

After change

  function f_TC_8_3_4_5_EUTRA() runs on EUTRA_PTC

  { //Inter-frequency E-UTRAN FDD - FDD / CSG Proximity Indication

    var template (value) CellPowerList_Type v_CellPowerList_T0, v_CellPowerList_T1, v_CellPowerList_T2, v_CellPowerList_T3,v_CellPowerList_T4, v_CellPowerList_T5;

    var CSG_Identity v_CSG_Identity_Cell23 := '000000000000000000000001011'B;

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var GutiParameters_Type v_GutiParams;

    var template (value) MobilityControlInfo v_MobilityControlInfo;

    var template (value) RadioResourceConfigDedicated v_RadioResourceConfigDedicated;

    var template (value) RachProcedureConfig_Type v_RachProcedureConfig;

    var template (value) DL_DCCH_Message v_RRCConnectionReconfiguration;

    var Frequency_IE_Type v_Frequency_IE_Cell23;

    timer t_Timer;

    v_CellPowerList_T0 := {

      cs_CellPower (eutra_Cell1, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_T1 := {

      cs_CellPower (eutra_Cell23, -79)

    };

    v_CellPowerList_T2 := {

      cs_CellPower (eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_ServingCellRS_EPRE),

      cs_CellPower (eutra_Cell23, tsc_NonSuitableOffCellRS_EPRE)

    };

    v_CellPowerList_T3 := {

      cs_CellPower (eutra_Cell1, tsc_ServingCellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE)

    };

    v_CellPowerList_T4 := {

      cs_CellPower (eutra_Cell23, -79)

    };

    // Init variables

    f_EUTRA_Init(c3);

    f_EUTRA_ChangeSysinfoCombination(eutra_Cell23, c7);

    // Set SysInfo for cell 23

    f_EUTRA_CellInfo_SetSysInfo_Csg_Ind(eutra_Cell23, true);

    f_EUTRA_CellInfo_SetSysInfo_Csg_Id(eutra_Cell23, v_CSG_Identity_Cell23);

    f_EUTRA_CellInfo_SetSIB4 (eutra_Cell23, cs_SIB4_AllParameters (omit, omit, cs_PhysCellIdRange(2, n4)));

    v_Frequency_IE_Cell23 := f_EUTRA_CellInfo_GetFrequencyIEs (eutra_Cell23);

    f_EUTRA_CellInfo_InitMaxReferencePower(eutra_Cell23, -79); 

    // Create and configure Cells

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell2);

    f_EUTRA_CellConfig_Def(eutra_Cell23);

    if (pc_Manual_CSG_Selection) {

      f_UT_USIM_Insert (UT, "USIM with service n86 available");

      // Empty UE Allowed CSG list in USIM

      f_EUTRA_Preamble_EmptyCSG_AllowedList (eutra_Cell23);

      // Broadcast correct CSG id

      f_EUTRA_CellInfo_SetSysInfo_Csg_Id(eutra_Cell23, v_CSG_Identity_Cell23);

      f_EUTRA_SS_ConfigureSysinfo(eutra_Cell23);

      // Add CSG23 in the Alllowed CSG list in USIM

      f_EUTRA_Manual_CSG_Selection( eutra_Cell23, v_CSG_Identity_Cell23);

    } else {

    // Insert a USIM with the CSG ID of Cell 23 in the UE's Allowed CSG list

      f_UT_USIM_Insert(UT, "The UE's Allowed CSG list contains the CSG ID of Cell 23");

    }
    f_EUTRA_SetCellPower (eutra_Cell23, tsc_NonSuitableOffCellRS_EPRE);

    f_EUTRA_TestBody_Set (true);

    // @siclog "Step 1" siclog@
...

    //@siclog "Step 76 - 83" siclog@

    //Steps 2 to 9 of the generic test procedure in TS 36.508 subclause 4.5.3.3 are performed on Cell 1.

    //NOTE: The UE performs the establishment of the new data radio bearer associated with the default EPS bearer context.

    f_EUTRA_RbEst_Def(eutra_Cell1);

    //@siclog "Step 84" siclog@

    //The SS adjusts cell levels according to row T4 of Table 8.3.4.5.3.2-1

    f_EUTRA_SetCellPowerList (v_CellPowerList_T4);
    //@siclog "Step 85 - 86" siclog@

    //The SS transmits an RRCConnectionReconfiguration message on Cell 1 including proximityIndicationEUTRA-r9 set to enabled.

    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,

                                                    cs_RRCConnectionReconfiguration_Common(v_RRC_TI,omit,omit,omit,omit,

                                                                                           cs_RRCConnectionReconfiguration_ReportProximityConfig_IEs(enabled)));

    //@siclog "Step 87" siclog@

    //Check: Does the UE transmit a ProximityIndication message with type-r9 set to "entering" proximity on Cell 1 within  [360]s?

    t_Timer.start(360.0);

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_508_ProximityIndication_Eutra(entering,v_Frequency_IE_Cell23.UL_DL_Earfcn.dl_CarrierFreq)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.3.4.5 Step 87");

    t_Timer.stop;

    f_EUTRA_TestBody_Set (false);

    // Postamble

    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED); //After removing the previous steps UE is on cell1 not cell2

  } // end of f_TC_8_3_4_5_EUTRA

4.2 Change 2
	Function name
	f_EUTRA_Authentication_InitNAS

	Reason for change
	KSIasme gets increased by one every time the Authentication procedure gets initialised again. The value of KSIasme equal to “111” in Authentication Request message is reserved. Therefore modulo of value 7 (111b) needs to be applied instead of current value 8.

	Summary of change
	Modulo 7 is now applied on the new value of KSIasme instead of modulo 8.

	TTCN module
	Common\EUTRA\EUTRA_SecurityFunctions.ttcn


Before change

    function f_EUTRA_Authentication_InitNAS(EUTRA_SecurityParams_Type p_Auth_Params,

                                          NAS_PlmnId p_PLMN,

                                          template (omit) Common_AuthenticationParams_Type p_Common_AuthenticationParams := omit) return EUTRA_SecurityParams_Type

  { /* @sic R5s100086 cl. 5.3.1 sic@ */

    var EUTRA_SecurityParams_Type v_Auth_Params := p_Auth_Params;

    if (isvalue(p_Common_AuthenticationParams)) {

      v_Auth_Params.AuthParams := valueof(p_Common_AuthenticationParams); //@sic R5s100182 sic@

    } else {

      v_Auth_Params.AuthParams := f_AuthenticationInit(v_Auth_Params.AuthParams); //@sic R5s100182 sic@

    }

    // Generates Ck, Ik, AUTN, XRES

    v_Auth_Params.Ks := v_Auth_Params.AuthParams.CK & v_Auth_Params.AuthParams.IK;

    // As per 33401 clause 6.2

    v_Auth_Params.KASME := fl_EUTRA_Authentication_S10 (v_Auth_Params, p_PLMN);

    // Generates KASME

    v_Auth_Params.KSIasme:= int2bit(((bit2int(v_Auth_Params.KSIasme)+1) mod 8), 3);

    // updates KSI

    v_Auth_Params.KSIsgsn := '111'B;

    // Invalidate KSIsgsn which is useful only in case of if KSIASME is derived from SGSN keys Ck and Ik.

    v_Auth_Params.NAS_Ciphering.K_NAS := fl_EUTRA_Authentication_S15 (tsc_NAS_Enc_Alg,v_Auth_Params.NAS_Ciphering.Algorithm,

                                                                      v_Auth_Params.KASME,

                                                                      v_Auth_Params.KDF);

    // Derives KNASenc key

    v_Auth_Params.NAS_Integrity.K_NAS := fl_EUTRA_Authentication_S15 (tsc_NAS_Int_Alg,v_Auth_Params.NAS_Integrity.Algorithm,

                                                                      v_Auth_Params.KASME,

                                                                      v_Auth_Params.KDF);

    // Derives KNASint key

    return v_Auth_Params;

  };
After change

     function f_EUTRA_Authentication_InitNAS(EUTRA_SecurityParams_Type p_Auth_Params,

                                          NAS_PlmnId p_PLMN,

                                          template (omit) Common_AuthenticationParams_Type p_Common_AuthenticationParams := omit) return EUTRA_SecurityParams_Type

  { /* @sic R5s100086 cl. 5.3.1 sic@ */

    var EUTRA_SecurityParams_Type v_Auth_Params := p_Auth_Params;

    if (isvalue(p_Common_AuthenticationParams)) {

      v_Auth_Params.AuthParams := valueof(p_Common_AuthenticationParams); //@sic R5s100182 sic@

    } else {

      v_Auth_Params.AuthParams := f_AuthenticationInit(v_Auth_Params.AuthParams); //@sic R5s100182 sic@

    }

    // Generates Ck, Ik, AUTN, XRES

    v_Auth_Params.Ks := v_Auth_Params.AuthParams.CK & v_Auth_Params.AuthParams.IK;

    // As per 33401 clause 6.2

    v_Auth_Params.KASME := fl_EUTRA_Authentication_S10 (v_Auth_Params, p_PLMN);

    // Generates KASME

    v_Auth_Params.KSIasme:= int2bit(((bit2int(v_Auth_Params.KSIasme)+1) mod 7), 3);

    // updates KSI

    v_Auth_Params.KSIsgsn := '111'B;

    // Invalidate KSIsgsn which is useful only in case of if KSIASME is derived from SGSN keys Ck and Ik.

    v_Auth_Params.NAS_Ciphering.K_NAS := fl_EUTRA_Authentication_S15 (tsc_NAS_Enc_Alg,v_Auth_Params.NAS_Ciphering.Algorithm,

                                                                      v_Auth_Params.KASME,

                                                                      v_Auth_Params.KDF);

    // Derives KNASenc key

    v_Auth_Params.NAS_Integrity.K_NAS := fl_EUTRA_Authentication_S15 (tsc_NAS_Int_Alg,v_Auth_Params.NAS_Integrity.Algorithm,

                                                                      v_Auth_Params.KASME,

                                                                      v_Auth_Params.KDF);

    // Derives KNASint key

    return v_Auth_Params;

  };                    
5) Execution Log Files

Qualcomm UE
The Qualcomm MSM8974 UE passed this test case on the Anritsu ME7832L LTE system with NAS and AS Snow 3G integrity and ciphering algorithm in TDD band 41. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file :

In the log file (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file

6 References

	[1]
	R5s140299: This archive comprises text format execution log file.

